
This chapter presents advanced randomization-based methods used in the analy-
sis of clinical endpoints. The nonparametric, randomization-based analysis of
covariance methodology, a versatile methodology that uses weighted least squares
to generate covariate-adjusted treatment effects under minimal assumptions,
is presented. These methods are implemented using a powerful SAS macro
(%NParCov4) that is applicable to a variety of clinical outcomes with the ability
to compute essentially-exact p-values and confidence intervals. Numerous clinical
trial examples are presented to illustrate the methodology and software.

2.1 Introduction

As explained in Chapter 1, adjustments for important covariates in the analysis of
clinical endpoints can be carried out using randomization-based and model-based
methods. The two classes of methods play complementary roles in the sense that
each method is useful within a particular setting, and the advantages of each
method offset the limitations of the other method. Chapter 1 deals with both the
model-based and basic randomization-based methods applied in clinical trials. This
chapter focuses on a class of advanced randomization-based methods with broad
applications to trials with different types of endpoints.

A detailed description of model-based approaches can be found in the beginning
of Chapter 1. This includes, for example, logistic regression models used in the
analysis of binary endpoints and the Cox proportional hazards model in settings
with time-to-event endpoints. An important feature of model-based methods is
that they rely on specific assumptions and thus can be quite sensitive to violations
from these assumptions. For example, considering proportional hazards models,
departures from the proportional hazards assumption are typically more evident for
baseline characteristics and measurements than for treatments. Also, proportional
hazards is applicable to treatments under the null hypothesis of equal treatment
effects. The issue of bias in a treatment comparison through departures from the
proportional hazards assumption for baseline characteristics and/or measurements
can be addressed by a randomization-based method that invokes randomization-
based covariance adjustment for the baseline characteristics and/or measurements
for an estimated hazard ratio from a proportional hazards model that only includes
treatments; see Saville and Koch (2013). Similar considerations apply to the propor-
tional odds model with respect to how to address departures from the proportional
odds assumption for baseline characteristics and/or measurements. See Tangen and
Koch (1999a). Additionally, the scope of randomization-based methods for ordinal
response includes Mann-Whitney rank measures of association. See Kawaguchi and
Koch (2011, 2015).

Nonparametric randomization-based analysis of covariance

It was pointed out in Chapter 1 that analysis of covariance serves two important
purposes in a randomized clinical trial. First, there is a reduction of variance for
the treatment estimate, which provides a more powerful statistical test and a more
precise confidence interval. Second, analysis of covariance provides an estimate
of the treatment effect, which is adjusted for random imbalances of covariates
between the treatment groups. Additionally, analysis of covariance models provide
estimates of covariate effects on the outcome (adjusted for treatment) and can test
for the homogeneity of the treatment effect within covariate-defined subgroups. The
nonparametric, randomization-based analysis of covariance method of Koch et al.
(1998) defines a versatile methodology that uses weighted least squares to generate
covariate-adjusted treatment effects with minimal assumptions. This methodology
is general in its applicability to a variety of outcomes, whether continuous, binary,
ordinal, incidence density, or time-to-event. (See Koch et al., 1998; Tangen and
Koch, 1999a, 1999b, 2000; LaVange and Koch, 2008; Moodie et al., 2010; Saville,
Herring, and Koch, 2010; Saville, LaVange, and Koch, 2010; Saville and Koch 2013;
and Zhao et al. 2016.) Further, its use has been illustrated in many clinical trial
settings, such as multi-center, dose-response, and non-inferiority trials (Koch and
Tangen, 1999; LaVange, Durham and Koch, 2005; and Tangen and Koch, 2001).
Similar covariate adjustment for randomized studies has been applied to multivariate
Mann-Whitney estimators of ordinal outcomes (Kawaguchi, Koch, and Wang, 2011;
and Kawaguchi and Koch, 2015).

The nonparametric, randomization-based analysis of covariance methodology
uses weighted least squares on the treatment differences of outcome and covari-
ate means. The resulting model estimates represent treatment effects for the
outcomes that have been adjusted for the accompanying covariates. Covariance-
adjustment in the treatment-effect estimates is a result of the assumed null difference
in covariate means, which, in turn, is a consequence of the underlying invoca-
tion of randomization to treatment. Full details of this methodology are left to
Appendix A.

This methodology has several advantages:

1. Applicability to a variety of outcomes, whether continuous, binary, ordinal,
incidence density, or time-to-event (See Appendices A.1 through A.5).

2. Minimal assumptions:

(a) Randomization to treatment is the principal structure necessary for hypoth-
esis testing.

(b) Confidence intervals of treatment estimates require the additional assump-
tion that subjects in the trial are a simple random sample of a larger
population.

(c) Survival analyses have the assumption of non-informative censoring.

3. Straightforward to accommodate stratification and to generate essentially-
exact p-values and confidence intervals (described in Appendices A.6 and A.7,
respectively).

4. Greater power of the adjusted treatment effect relative to the unadjusted.

However, there are some disadvantages:

1. No estimates for the effects of covariates or strata are available.

2. No direct way to test homogeneity of treatment effects within subgroups that
are defined by the covariates.

3. Results are not directly generalizable to subjects with covariate distributions
that differ from those in the trial.

4. Is not appropriate when the same multivariate distribution of the covariates
cannot be assumed to apply to both treatment groups.

It is important to note that nonparametric, randomization-based analysis of
covariance is not intended to replace likelihood-based methodologies such as general-
ized linear models (see Chapter 1). On the contrary, this methodology is expected to
be used in a complementary fashion with model-based methods, as long as patients
are randomized to treatment (see Tangen and Koch, 1999a, 2000; and Saville and
Koch, 2013, as examples). In a regulatory setting, randomization-based, nonpara-
metric analysis of covariance can establish whether a treatment effect exists under
minimal statistical assumptions. Once a treatment effect has been identified, other
scientific questions can be explored under some additional plausible assumptions
using more traditional models. The examples in this chapter focus on the analysis
and interpretation of data using nonparametric, randomization-based analysis of
covariance. Additional analyses to address other scientific questions are not shown.
(Interested readers might choose to use the model-based methods introduced in
Chapter 1 to compare with the findings presented here.)

For example, as outlined in Tangen and Koch (1999a), the specification of
statistical methods in a protocol or analysis plan is typically required prior to the
unblinding of study data. It is recommended that randomization-based analysis of
covariance be specified as the primary analysis because of minimal assumptions and
the availability of essentially-exact p-values. Model-based methods, such as a logistic
regression model in the case of binary outcomes, can be used in a supportive fashion
to assess the effects of covariates, potential interactions of treatment with covariates,
and the generalizability of findings to populations not represented by the subjects in
the current study. This last point is particularly important. For nonlinear models,
such as logistic regression, the estimated treatment effect is specific to a population
of patients with a specific set of values for the covariates. Randomization-based,
nonparametric analysis of covariance estimates and tests the treatment effect with
adjustment for covariates for the randomized population of patients.

This chapter provides a detailed description of the nonparametric, randomization-
based analysis of covariance methodology. A description of key features of the
advanced randomization-based framework is provided in Section 2.1. Case studies
used throughout this chapter are presented in Section 2.2. Section 2.3 introduces
the %NParCov4 macro that performs these analyses with the ability to compute
essentially exact p-values and confidence intervals. We discuss the strengths and
limitations of the randomization-based methodology and provide complementary
analysis strategies using the case studies in Sections 2.4 through 2.9. As stated above,
technical details are left to the Appendix. Appendix A summarizes the theoretical
details of these methods while Appendix B documents the %NParCov4 macro.

The SAS code and data sets included in this chapter are available on the book’s
website at http://support.sas.com/publishing/authors/dmitrienko.html.

2.2 Case studies

The following four case studies, based on real clinical trials, will be used in this
chapter to illustrate advanced randomization-based methods for the analysis of
clinical endpoints.

2.2.1 Case study 1 (Respiratory disorder trial)

The goal of this randomized clinical trial was to compare an active treatment versus
placebo for 111 subjects with a respiratory disorder at two centers (Koch et al.,
1990; and Stokes, Davis, and Koch, 2012). The primary endpoint was ordinal, and
subjects were classified with an overall score of 0, 1, 2, 3, or 4 (terrible, poor, fair,
good, or excellent) at each of four visits (V1 through V4). Covariates included a
baseline score, gender, and age (in years) at study entry. Treatment was coded 1 for
active treatment, 0 for placebo, while gender was coded 1 for male, and 0 for female.
In addition to the ordinal scale, dichotomous outcomes, such as (good, excellent)
versus (terrible, poor, fair) with values of 1 or 0, respectively, were of interest. The
first few rows of the respiratory data are presented in Table 2.1.

TABLE 2.1 Respiratory disorder data

Center Treatment Gender Age Baseline V1 V2 V3 V4

0 0 1 46 2 2 2 2 2
0 0 1 28 2 0 0 0 0
0 0 1 44 3 4 3 4 2
0 0 0 13 4 4 4 4 4
0 0 1 43 2 3 2 4 4
0 0 1 37 3 2 3 3 2
0 0 1 23 3 3 1 1 1
...

...
...

...
...

...
...

...
...

There are several ways to analyze the overall score from this clinical trial.
For example, if the distribution of overall scores at a visit appears to have an
approximately normal shape, a linear model might be appropriate to assess the
effect of treatment with adjustment for covariates (Section 2.4). Alternatively, the
clinical team might prefer to compare the proportion of patients with good or
excellent response between the two treatments. Here, a logistic regression model
can be used to model the effect of covariates on treatment response (Section 2.5).
Finally, if the proportional odds assumption is appropriate, a proportional odds

model can be fit to the overall score with adjustment for covariates (Section 2.6).
The proportional odds assumption implies that the model coefficients for all binary
splits of the outcome (0 vs >0, ≤1 vs >1, ≤2 vs >2, and ≤3 vs 4) are identical.

2.2.2 Case study 2 (Osteoporosis trial)

Data from a randomized clinical trial in 214 postmenopausal women with osteoporosis
at two study centers were used to evaluate the effectiveness of a new drug (t) versus
placebo (p) (Stokes et al., 2012). Both treatment groups had access to calcium
supplements, received nutritional counseling, and were encouraged to exercise
regularly. The number of fractures experienced each year were recorded through
the trial’s duration (three years). Seven women discontinued during the third year
and are considered to have a 6-month exposure for this year. Otherwise, yearly
exposure is considered to be 12 months. The outcome of interest was the rate of
fracture per year. Adjusting for age as a covariate and for centers as strata would
be appropriate.

Table 2.2 contains the first few rows of the osteoporosis data set. Year 1, Year
2, and Year 3 contain the number of fractures experienced in Years 1, 2, and 3,
respectively. Total is the total number of fractures experienced over the duration of
the trial. Exposure equals 3 years save for the 7 patients who discontinued early with
2.5 years of exposure. Yearly risk is defined as the total number of fractures (Total)
divided by Exposure. Center A and B were recoded as 1 and 2, while Treatment p
and t were recoded as 1 and 2 for the analysis. Yearly risk can be analyzed using
a Poisson regression model to assess the effect of treatment with adjustment for
covariates (Section 2.7).

TABLE 2.2 Osteoporosis data

Age Center Treatment Year 1 Year 2 Year 3 Total Exposure

56 A p 0 0 0 0 3
71 A p 1 0 0 1 3
60 A p 0 0 1 1 3
71 A p 0 1 0 1 3
78 A p 0 0 0 0 3
67 A p 0 0 0 0 3
49 A p 0 0 0 0 3
...

...
...

...
...

...
...

...

2.2.3 Case study 3 (Aneurysmal subarachnoid hemorrhage trial)

Nicardipine is indicated for the treatment of high blood pressure and angina in
both oral and intravenous formulations. Data from 902 subjects from a randomized
,two-arm intravenous study of high dose nicardipine was presented in Haley, Kassell,
and Torner (1993). The primary endpoint was improvement in patient recovery
according to the Glasgow Outcome Scale, with the incidence of cerebral vasospasm
and the incidence of death or disability due to vasospasm serving as important
secondary endpoints (Jennett and Bond, 1975). Subjects were treated with drug or
placebo up to 14 days after the hemorrhage occurred. Here, we compare the number
of treatment-emergent, serious adverse events between nicardipine (Treatment = 1)
and placebo (Treatment = 0), accounting for the varying durations of treatment
exposure. Age in years and gender are available as covariates.

Table 2.3 contains the first few rows of the aneurysmal, subarachnoid hemorrhage
data set. Count is the total number of treatment-emergent, serious adverse events

that occur over the duration of trial exposure (Days). Male is coded 1 if the patient
is a male, and 0 if the patient is a female. The number of adverse events can be
analyzed using a Poisson regression model using exposure as an offset to assess the
effect of treatment with adjustment for covariates (Section 2.8).

TABLE 2.3 Aneurysmal, subarachnoid hemorrhage Data

Treatment Count Days Age Male

0 0 13 63 0
1 0 11 66 1
0 0 10 31 0
1 6 1 48 0
0 1 13 67 0
0 0 11 32 1
1 5 3 63 1
...

...
...

...
...

2.2.4 Case study 4 (Colorectal cancer trial)

Data from a randomized study of 110 subjects with colorectal cancer are presented
to compare an experimental treatment to a control (Table 2.4). Treatment (Trt) is
coded 1 or 0 for active treatment or control, respectively. Data for two endpoints
are available: the time until death and the time until disease progression. Disease
progression was defined using standardized RECIST criteria (Eisenhauer et al.,
2009). Subjects who did not experience death or disease progression were censored
on their last day of study participation. Overall survival (OS) and progression-free
survival (PFS) are each represented by two variables. The variable OS Time is the
number of days from randomization until death. OS Event equals 1 if the patient
died. Otherwise, a value of 0 indicates that the patient was censored on their last
day of study participation. For PFS, the variable PFS Time is the number of days
from randomization until disease progression. PFS Event equals 1 if the patient
experienced disease progression. Otherwise, a value of 0 indicates that the patient
was censored on their last day of study participation.

TABLE 2.4 Colorectal cancer data

Trt OS Time OS Event PFS Time PFS Event KRAS Days Length

1 75 1 40 1 WT 805 121
1 406 0 68 1 WT 3186 160
1 225 1 118 1 M 2449 101
1 89 1 36 1 M 457 97
1 309 1 72 1 WT 1396 158
0 123 1 50 1 WT 2341 120
1 465 0 48 1 WT 1513 64
...

...
...

...
...

...
...

...

Covariates include the number of days from initial cancer diagnosis to start of
study drug administration (Days), and the sum of the longest lesion diameters in
millimeters at baseline (Length). A flag as to whether the KRAS gene was wild-type
(WT) or mutated (M) can be used as a stratification variable. For the analysis,
a numeric variable mutant will be defined with values of 0 or 1 to indicate wild-
type or mutated, respectively. Time-to-event endpoints are often analyzed using
non-parametric Kaplan-Meier analysis (Section 2.9). Alternatively, Cox proportional-
hazards regression is used when it is important to adjust for one or more covariates.

2.3 %NParCov4 macro

Analysis of clinical endpoints based on the nonparametric, randomization-based
methodology is easily performed using the %NParCov4 macro, a SAS/IML macro
that supports functionality for many of the analysis methods described to date,
with the ability to compute essentially-exact p-values and confidence intervals. This
macro will be used in Sections 2.4 through 2.9 to perform adjusted analyses of
continuous, categorical, count-type, and time-to-event endpoints.

A macro call including all options is presented in Program 2.1.

PROGRAM 2.1 General %NParCov4 Macro Call

%NPARCOV4(OUTCOMES = , COVARS = , EXPOSURES = , TRTGRPS = , HYPOTH = ,

TRANSFORM = , STRATA = , COMBINE = , C = , ALPHA = , EXACT = , SEED = ,

NREPS = , SYMSIZE = , DETAILS = , DSNIN = , DSNOUT = );

The user can specify one or more numeric outcomes (OUTCOMES), one or more
numeric covariates, or none if no covariate adjustment is required (COVARS). For
time-to-event outcomes or for analyses of incidence densities, the user specifies one
or more numeric exposure variables (EXPOSURES). A two-level numeric variable is
required to define treatment (TRTGRPS).

Hypothesis (HYPOTH) is specified as either NULL or ALT and impacts whether
the covariance matrices of responses and covariates in each strata are computed
across treatments (under the null) or within treatments (under the alternative).
HYPOTH = NULL by default. In general, the variance under the null would be the
principal structure for producing p-values for hypothesis testing under the null
when the null corresponded to no difference between treatments (or more formally,
the strong null that each patient would have the same outcomes regardless of the
assigned treatment). The variance under the alternative will be used for computing
confidence intervals. Here, the standard errors address replicated random samples
from a population for which the randomized population is arguably a random
sample. P -values under the alternative might be useful since they will agree with
the confidence interval, at least when both are based on normal approximations.
Throughout this chapter, we take the approach that p-values will be computed under
the null, with corresponding confidence intervals computed under the alternative
hypothesis. The user can specify α (ALPHA) to determine the confidence-level of the
100(1− α)% confidence intervals. By default, ALPHA = 0.05.

A single numeric variable can be provided to indicate strata (STRATA), and there
are options to specify when covariate adjustment occurs relative to accounting for
strata (COMBINE). By default, COMBINE = NONE, which assumes no stratification.
When strata are present, the recommendation for when to account for covariate
adjustment is typically determined by sample size. See Appendix A.6 for guidance
on appropriate choices for COMBINE. How the estimates from the strata are combined
is determined by C, which is the exponent for a Mantel-Haenszel weight. When C

= 1 (default), Mantel-Haenszel weights are used to combine strata. This considers
the sample size of each stratum when combining estimates. When C = 0, strata are
weighted equally, effectively ignoring the size of each stratum. Values of C between
0 or 1 put lesser or greater emphasis on strata sample sizes, respectively.

For computing resampling-based exact analyses (EXACT, NO by default), the user
can specify a seed (SEED, 0 by default) to generate reproducible analyses, the number
of simulations to conduct (NREPS), and the amount of memory to allocate to the
symbol space (SYMSIZE, 20000 by default). Whether macro details are written to
the SAS log or not is controlled by the DETAILS option (YES by default). For lengthy
simulations, the user should consider setting DETAILS = NO; specifying OPTIONS

NONOTES; and using a SAS macro variable to report simulation progress. See Program
2.18 for an example.

The input data set name (DSNIN) specifies a single data set containing analysis
data. The prefix for output data sets (DSNOUT) enables the user to customize the
names of the output data sets generated by %NParCov4; see Appendix B.1.15 for a
complete list of output data sets.

Appendix B.1 provides details for all %NParCov4 macro options, and Appendix
B.2 describes the contents of the generated output data sets.

2.4 Analysis of ordinal endpoints using a linear model

This section and the next five sections provide illustrations of the nonparametric,
randomization-based methodology by applying it to the four clinical trial examples
defined earlier in the chapter.

This section applies the randomization-based methodology to the analysis of an
ordinal endpoint in Case study 1. Suppose that the trial’s sponsor wants to compare
the average response for the ordinal respiratory outcome at Visit 1 between the
active and placebo treatments with adjustment for gender, age, and baseline score.
We can perform a stratified analysis using centers as strata, weighting each center
using Mantel-Haenszel weights (C = 1). Further, we combine treatment estimates
across strata prior to covariance adjustment using COMBINE = FIRST, assuming
that the sample size per stratum is small. The %NParCov4 call for this example is
provided in Program 2.2.

PROGRAM 2.2 Linear Model with Adjustment for Covariates for Visit 1

%NParCov4(OUTCOMES = V1, COVARS = gender age baseline, C=1,

HYPOTH = NULL, STRATA = center, TRTGRPS = treatment, TRANSFORM = NONE,

COMBINE = FIRST, EXACT = YES, NREPS = 5000, SEED = 36, DSNIN = RESP,

DSNOUT = OUTDAT);

In Program 2.2, HYPOTH = NULL so that we can first obtain p-values of interest
for our analysis. The option DSNOUT = OUTDAT sets the value OUTDAT as the prefix
for the output data sets. Throughout this chapter, we assume DSNOUT = OUTDAT

and C = 1. We refrain from repeating these options in the programs below unless
the values change. For this particular example, the following data sets are created:

1. OUTDAT BETASAMP, which contains the observed and NREPS = 5000 permutation-
based estimates of the treatment effect.

2. OUTDAT COVBETA, which contains the covariance matrix of the covariate-adjusted
treatment effects for the outcomes.

3. OUTDAT COVTEST, which contains the criterion for covariate imbalance.

4. OUTDAT DEPTEST, which contains the treatment estimate, its standard error, and
a two-sided asymptotic p-value.

5. OUTDAT EXACT, which contains one- and two-sided p-values for an essentially-
exact analysis of NREPS = 5000 permutation samples using SEED = 36.

The data set OUTDAT COVTEST contains the criterion for covariate imbalance
Q=6.46, which, with sufficient sample size, is distributed χ2

(3). The p-value for the
criterion is 0.0911. While the exact p-value based on 5000 permutation samples is
0.0920, which is not contrary to random imbalance for the distributions of covariates

between the two treatments. In other words, randomization provided statistically
similar distributions of covariates across the strata.

This criterion should not be considered a test for validity of nonparametric,
randomization-based analysis of covariance, but merely as a measure of the degree
of imbalance among covariates between the treatments. Due to the randomization
requirement of the methodology, significant imbalance detected among the covariates
should be interpreted as a chance finding. Adjusting for the covariates in the analysis
removes this imbalance; the results of an unadjusted analysis would be confounded
with the imbalance among the covariates. Small p-values for the covariate imbalance
criterion are more appropriately referred to as ‘atypical relative to ”chance” rather
than ”significant.”

Data set OUTDAT DEPTEST summarizes results for the treatment difference that
is provided in Table 2.5. Outcome, Treatment Effect, Standard Error, Test Statistic,
and P-value are represented by the OUTDAT DEPTEST variables OUTCOMES, BETA,
SEBETA, Q J, and PVALUE.

TABLE 2.5 Linear Model with Adjustment for Covariates for Visit 1

Outcome Treatment Effect Standard Error Test Statistic P-value

Visit 1 0.4008 0.1714 5.4690 0.0194

An exact p-value based on 5000 permutation data sets is 0.0162. If there was
interest in confidence intervals, we could specify HYPOTH = ALT in Program 2.2
above. An additional data set, OUTDAT CI, is created. It contains asymptotic 95%
confidence intervals. Exact 95% bias-corrected and accelerated (BCa) and percentile
intervals are included in the OUTDAT EXACT data set. The asymptotic 95% confidence
interval for the treatment estimate is (0.1001, 0.7531). The corresponding bootstrap
BCa interval is (0.0974, 0.7749), and the percentile interval is (0.0901, 0.7646) based
on 5000 bootstrap data sets.

The treatment effect in Table 2.5 represents the increase in the overall score for
visit 1 for the active treatment compared to placebo. Based on our above analysis
for Visit 1, we see that there is a significant treatment effect when stratifying by
center and adjusting for gender, age, and baseline score. The treatment estimate is
positive, indicating that the active drug had a greater positive effect, by about 0.40
on average, than placebo. The treatment estimate of approximately 0.40 could also
be interpreted as corresponding to a 0.40/4 = 0.10 difference between treatments
for the average of the cumulative probabilities of a score of at least 1, at least 2, at
least 3, and at least 4. In contrast, the model without adjustment for covariates
(Program 2.3) provides the results in Table 2.6.

PROGRAM 2.3 Linear Model without Adjustment for Covariates for Visit 1

%NParCov4(OUTCOMES = V1, COVARS =, HYPOTH = NULL, STRATA = center,

TRTGRPS = treatment, TRANSFORM = NONE, COMBINE = FIRST, EXACT = YES,

NREPS = 5000, SEED = 36, DSNIN = RESP);

TABLE 2.6 Linear Model without Adjustment for Covariates for Visit 1

Outcome Treatment Effect Standard Error Test Statistic P-value

Visit 1 0.3935 0.2032 3.7497 0.0528

An exact p-value based on 5000 permutation data sets is 0.0542. For confidence
intervals, we could specify HYPOTH = ALT in Program 2.3 above. The asymptotic
95% confidence interval for the treatment estimate is (0.0024, 0.7846). Here, the
BCa and percentile intervals are (0.0197, 0.7906) and (0.0131, 0.7799), respectively,
based on 5000 bootstrap data sets.

Note that for this particular example, the interpretation of the p-value and the
confidence intervals don’t coincide. This phenomenon does occur from time to time.
This is because the p-value is based on re-randomizations of the population, while
the confidence interval is based on repeated random sampling. This phenomenon
can also occur for 2 × 2 tables where the p-value for Fisher’s exact test exceeds
0.05, but a confidence interval for the risk difference excludes 0. In this setting,
Fisher’s exact test is a randomization method based on re-randomizations of the
population. But the confidence interval for the risk difference is based on repeated
random sampling from a binomial distribution. The good news is that the result for
the p-value and the confidence interval agree most of the time in terms of indicating
whether a treatment effect exists. Otherwise, when the p-values and confidence
intervals provide conflicting information, the p-value from the randomization test is
based on more realistic and minimal assumptions, and should take precedence.

It is clear from the results of Tables 2.5 and 2.6 that adjustment for covariates
provides a more sensitive comparison between the two treatments. To further
illustrate the importance of covariate adjustment and the power of the nonparamet-
ric, randomization-based analysis of covariance methodology, Table 2.7 presents
asymptotic 95% confidence intervals and interval widths for unstratified analyses
similar to those presented in Programs 2.2 and 2.3, alongside comparable para-
metric analyses generated using PROC MIXED (Program 2.4). Intervals for both
methods are non-significant or significant for the unadjusted or adjusted analysis,
respectively. Further, in both the adjusted and unadjusted case, nonparametric,
randomization-based analysis of covariance provides a narrower confidence interval
than the model-based analysis, with greater precision observed when adjusting for
covariates.

TABLE 2.7 Estimated Treatment Effects and Asymptotic 95% Confidence Intervals by Method

Method Estimate Lower Limit Upper Limit Width

Regression: Unadjusted 0.3996 -0.0129 0.8122 0.8251
NParCov: Unadjusted 0.3935 0.0024 0.7846 0.7822
Regression: Adjusted 0.4456 0.0995 0.7918 0.6923
NParCov: Adjusted 0.4266 0.1001 0.7531 0.6530

PROGRAM 2.4 Parametric Analyses with and without Adjustment for Covariates for Visit 1

proc mixed data = resp;

ods output estimates = mixedc;

model v1 = gender age baseline treatment;

estimate "Adjusted Treatment Effect" treatment 1;

run;

proc mixed data = resp;

ods output estimates = mixed;

model v1 = treatment;

estimate "Unadjusted Treatment Effect" treatment 1;

run;

Suppose there is interest in examining the treatment effect across all 4 visits
while adjusting for covariates and accounting for strata (Program 2.5).

PROGRAM 2.5 Linear Model with Adjustment for Covariates for All Visits

%NParCov4(OUTCOMES = V1 V2 V3 V4, HYPOTH = NULL,

COVARS = gender age baseline, STRATA = center, TRANSFORM = NONE,

TRTGRPS = treatment, COMBINE = FIRST, DSNIN = RESP);

The output is provided in Table 2.8.

TABLE 2.8 Linear Model with Adjustment for Covariates for All Visits

Outcome Treatment Effect Standard Error Test Statistic P-value

Visit 1 0.4008 0.1714 5.4690 0.0194
Visit 2 0.9516 0.2213 18.4901 < 0.0001
Visit 3 0.8160 0.2386 11.6948 0.0006
Visit 4 0.6175 0.2377 6.7513 0.0094

The treatment estimates from OUTDAT DEPTEST and the covariance matrix from
the data set OUTDAT COVBETA can be used to test more complex hypotheses. We
could test the global hypothesis of a non-null treatment effect at the 4 visits,

H0 : f1 = 0, f2 = 0, f3 = 0, f4 = 0,

where fi is the treatment effect at the ith visit. The test statistic

QGlobal = β̂>C>(CVβ̂C
>)−1Cβ̂

is distributed χ2
(4), where C = I4 (an identity matrix of order four), and Vβ̂ is the

covariance matrix of the covariate-adjusted treatment effects of visits 1 to 4 for the
overall response. The SAS/IML code in Program 2.6 performs the analysis.

PROGRAM 2.6 Global Test for Treatment Response Across Visits

proc iml;

use _OUTDAT_DEPTEST;

read all var{beta} into B;

close;

use _OUTDAT_COVBETA;

read all var{V1 V2 V3 V4} into V;

close;

C = I(4);

Q = t(B) * t(C) * inv(C * V * t(C)) * C * B;

p = 1 - probchi(Q, nrow(C));

print Q p;

quit;

With QGlobal = 19.44 and p = 0.0006, the analysis suggests that a non-null treatment
effect exists for at least one visit. Note that the 4 degree-of-freedom test QGlobal
would tend to have much less power than the 1 degree-of-freedom test that uses the
average of the treatment differences across the visits; this would also be considered
a global test of a treatment effect across the 4 visits.

An additional analysis can test the null hypothesis that the treatment effect does
not differ across study visits,

H0 : f1 = f2 = f3 = f4.

Here, the result of the test statistic and p-value for treatment × visit should be
based on the variance under the alternative rather than the variance under the null,
since the treatment effects are already understood to be non-null. In this case, the
test statistic

QV isit = β̂>C>(CVβ̂C
>)−1Cβ̂

is distributed χ2
(3), where

C =

 1 0 0 −1
0 1 0 −1
0 0 1 −1

 .

Changing HYPOTH = ALT in Program 2.5 and setting

C = {1 0 0 -1, 0 1 0 -1, 0 0 1 -1};

in Program 2.6 provides QTreatment×V isit= 12.57 and p = 0.0057. This suggests
that the treatment effect differs across visits. In Section 2.16, we demonstrate how
to generate essentially exact results for analyses with more than one response.

2.5 Analysis of binary endpoints

Case study 1 will be used to illustrate the analysis of binary endpoints with appro-
priate covariate adjustment. Specifically, in this section, we will analyze the binary
response of (good, excellent) versus (terrible, poor, fair) at Visit 1. If the primary
interest is in the test of the treatment effect, the most appropriate covariance matrix
is the one that applies to the randomization distribution under the null hypothesis of
no treatment difference. An exact p-value can be obtained; under the null hypothesis,
all resampled data sets have the same covariance matrix. For illustration, we obtain
weighted estimates across strata prior to covariance adjustment. The %NParCov4

call for this example is provided in Program 2.7.

PROGRAM 2.7 Difference in Proportions with Adjustment for Covariates for Visit 1

data RESP;

set RESP;

v1goodex = (V1 in (3,4));

run;

%NParCov4(OUTCOMES = v1goodex, COVARS = gender age baseline,

HYPOTH = NULL, STRATA = center, TRTGRPS = treatment, TRANSFORM = NONE,

COMBINE = FIRST, EXACT = YES, NREPS = 5000, SEED = 78, DSNIN = RESP);

The criterion for covariate imbalance Q = 6.46 is distributed χ2
(3). The asymptotic

p-value for the criterion is 0.0911, and the exact p-value provided in OUTDAT COVTEST

based on 5000 permutation samples is 0.0922. It appears that randomization provided
reasonably equal distributions of covariates across the strata. With TRANSFORM =

NONE, the model provides the treatment estimate for the difference in proportions
in Table 2.9.

TABLE 2.9 Linear Model with Adjustment for Covariates for Visit 1

Outcome Treatment Effect Standard Error Test Statistic P-value
Visit 1 0.1839 0.0781 5.5455 0.0185

The exact p-value for the treatment effect provided in OUTDAT EXACT, based
on 5000 permutation samples, is 0.0164. Therefore, active treatment provides a
higher proportion of good or excellent responses when adjusting for gender, age, and
baseline score. If there was interest in confidence intervals for the odds ratio, we
could specify HYPOTH = ALT and TRANSFORM = LOGISTIC in the macro call above.
However, this analysis causes an error for %NParCov4! What went wrong? The
outcome was rare enough that some of the resampled data sets had no successes
for at least one treatment arm within strata, and this caused the estimation to fail.
If the outcome is expected to be rare, caution should be taken in specifying the
analysis. For example, the following analysis adjusting for center as a covariate
generates exact results (Program 2.8).

PROGRAM 2.8 Logistic Model with the Addition of Center as a Covariate for Visit 1

%NParCov4(OUTCOMES = v1goodex, COVARS = center gender age baseline,

HYPOTH = ALT, STRATA = NONE, TRTGRPS = treatment, TRANSFORM = LOGISTIC,

EXACT = YES, NREPS = 5000, SEED = 78, DSNIN = RESP);

The data set OUTDAT CI is created for all models and contains asymptotic con-
fidence intervals for the treatment parameters when HYPOTH = ALT. However, an
additional data set OUTDAT RATIOCI is produced here. It contains the asymptotic
confidence intervals for the odds ratios. The odds ratio and asymptotic 95% con-
fidence interval obtained from the data set OUTDAT RATIOCI is 2.2707 (1.2086,
4.2665). The BCa and percentile intervals for the odds ratio are (1.1510, 4.6950)
and (1.1681, 4.7672), respectively, based on 5000 samples.

Ideally, the first analysis should be regenerated using the new options (center as
a covariate) to be consistent with the analysis from Program 2.8. The appropriate
code is presented in Program 2.9 (results not shown).

PROGRAM 2.9 Difference in Proportions with Center as a Covariate for Visit 1

%NParCov4(OUTCOMES = v1goodex, COVARS = center gender age baseline,

HYPOTH = ALT, STRATA = NONE, TRTGRPS = treatment, TRANSFORM = NONE,

EXACT = YES, NREPS = 5000, SEED = 78, DSNIN = RESP);

2.6 Analysis of ordinal endpoints using a proportional odds model

Case study 1 can also be used to illustrate an analysis strategy based on a proportional
odds model. Rather than dichotomize the ordinal response in this clinical trial, we
can analyze the outcome using a proportional odds model. There are few responses
for poor and terrible so these categories will be combined for analysis. We can define
3 binary indicators to indicate excellent response or not, at least good response or
not, or at least fair response or not (Program 2.10).

PROGRAM 2.10 Proportional Odds Model with Adjustment for Covariates for Visit 1

data RESP;

set RESP;

v1ex = (V1 in (4));

v1goodex = (V1 in (3,4));

v1fairgoodex = (V1 in (2,3,4));

run;

%NParCov4(OUTCOMES = v1ex v1goodex v1fairgoodex, COVARS = gender age baseline,

HYPOTH = NULL, STRATA = center, TRTGRPS = treatment, TRANSFORM = PODDS,

COMBINE = FIRST, DSNIN = RESP);

A model where TRANSFORM = PODDS generates an additional data set OUTDAT HOMOGEN

that contains a test for the appropriateness of the proportional odds assumption.
The test of proportional odds is Q = 3.69, which is distributed χ2

(2) and provides
a p-value of 0.1578. Thus, we do not reject the assumption of proportional odds.
Further, the covariate imbalance criterion, which in this case is a joint test of
proportional odds and covariate imbalance (for gender, age, and baseline), is
borderline (p=0.0709). The model provides the following treatment estimate of the
log odds ratio comparing active treatment to placebo in Table 2.10.

TABLE 2.10 Proportional Odds Model with Adjustment for Covariates for Visit 1

Outcome Treatment Effect Standard Error Test Statistic P-value

Visit 1 0.6233 0.3046 4.1857 0.0408

When HYPOTH = ALT is changed in Program 2.10, the data set OUTDAT RATIOCI

contains odds ratio and asymptotic 95% confidence interval of 1.9548 (1.0455,
3.6548). Therefore, the odds of a positive response in the test treatment are greater
than for placebo. Finally, had the test of homogeneity been statistically significant,
the proportional odds assumption would not have applied for this analysis. In this
case, the full model could be obtained by using TRANSFORM = LOGISTIC in the
macro call above to estimate separate odds ratios for each binary outcome. See
Program 2.11 for a sample program (results not shown).

PROGRAM 2.11 Model with Adjustment for Covariates for Visit 1 - Proportional Odds Does Not
Apply

%NParCov4(OUTCOMES = v1ex v1goodex v1fairgoodex, COVARS = gender age baseline,

HYPOTH = ALT, STRATA = center, TRTGRPS = treatment, TRANSFORM = LOGISTIC,

COMBINE = FIRST, DSNIN = RESP);

2.7 Analysis of continuous endpoints using the log-ratio of two means

Analysis of continuous endpoints in the context of the nonparametric, randomization-
based methodology can be illustrated using the osteoporosis trial from Case study 2.
The analysis focuses on comparing the yearly risk of fractures between the two
treatments using a risk ratio. We calculate the yearly risk for each subject as the
total number of fractures divided by 3 years (or 2.5 years for the 7 subjects who
discontinued early). Treatment (active = 2, placebo = 1) and center are recoded
to be numeric. For illustration, we take weighted averages across the strata prior
to taking the log transformation (based on the natural logarithm) and applying
covariance adjustment (Program 2.12).

PROGRAM 2.12 Log-Ratio of Two Means with Adjustment for Covariates

%NParCov4(OUTCOMES = YEARLYRISK, COVARS = age, HYPOTH = NULL,

STRATA = center, TRTGRPS = treatment, TRANSFORM = LOGRATIO,

COMBINE = PRETRANSFORM, EXACT = YES, NREPS = 5000, SEED = 22,

DSNIN = FRACTURE);

The covariate imbalance criterion for this example is Q = 0.0536 that is dis-
tributed χ2

(1) with p = 0.8170. The exact p-value is 0.8286. It appears that random-
ization provided a reasonably equal distribution of age between treatment groups.
The model generates the treatment estimate of the log-ratio of the two means. It is
presented in Table 2.11.

TABLE 2.11 Log-Ratio of Two Means with Adjustment for Covariates

Outcome Treatment Effect Standard Error Test Statistic P-value

Yearly Risk -0.5803 0.3006 3.7259 0.0536

The exact p-value is 0.0626 based upon 5000 samples. If there was interest in
confidence intervals for the log-ratio of the means, we could specify HYPOTH =

ALT in the macro call above. Exponentiating the treatment estimate provides the

ratio 0.5600 which has 95% confidence interval (0.3068, 1.0222). Despite fractures
being (1/0.56 − 1) × 100% = 79% more likely per year for subjects on placebo
compared to active drug, this difference is of borderline significance when adjusting
for age. The BCa and percentile intervals for the risk ratio are (0.2823, 0.9999)
and (0.2882, 1.0209), respectively, based on 5000 samples. Here, the BCa intervals
imply a significant difference between placebo and active drug that conflicts with
the asymptotic and percentile intervals. Doubling the number of bootstrap samples
provides a BCa interval that includes 1. Note that had there been greater variability
in the duration of exposure for each subject, we could have analyzed with the
methods in the following section.

2.8 Analysis of count endpoints using log-incidence density ratios

Several of the 40 clinical sites in the aneurysmal subarachnoid hemorrhage trial
from Case study 3 had low enrollment, with fewer than two patients per treatment
in some cases. In order to analyze under the alternative hypothesis, we will ignore
center as a stratification variable. For each treatment, the incidence density will be
the number of treatment-emergent, serious adverse events that a subject experienced,
normalized by the number of days that he or she was in the trial. The ratio of
the incidence densities (nicardipine over placebo) will be treated as the primary
endpoint in the trial. Treatment estimates are adjusted for age and gender in the
macro call of Program 2.13.

PROGRAM 2.13 Log-Incidence Density Ratios with Adjustment for Covariates

%NParCov4(OUTCOMES = count, COVARS = age male, EXPOSURES = days,

HYPOTH = NULL, TRTGRPS = treatment, TRANSFORM = INCDENS,

EXACT = YES, NREPS = 5000, SEED = 77, DSNIN = nic);

The covariate imbalance criterion for this example is Q = 0.81, which is dis-
tributed χ2

(2) with the p-value of 0.666. The exact p-value is 0.6810. It appears
that randomization provided a reasonably equal distribution of age and gender
between treatment groups. The treatment estimate of the log-incidence density ratio
is presented in Table 2.12.

TABLE 2.12 Log-Incidence Density Ratios with Adjustment for Covariates

Outcome Treatment Effect Standard Error Test Statistic P-value

Incidence Density -0.0718 0.1428 0.2526 0.6153

The exact p-value is 0.6104 based on 5000 samples. When HYPOTH = ALT is
changed in Program 2.13, the data set OUTDAT RATIOCI contains incidence density
ratio 0.9307 which has 95% confidence interval (0.7007, 1.2363). Treatment-emergent,
serious adverse events are (1/0.9307 - 1) × 100% = 7.4% more likely per day for
subjects on placebo compared to nicardipine. This difference is not statistically
significant when adjusting for the covariates. The BCa and percentile intervals for
the incidence density ratio are (0.6973, 1.2228) and (0.7009, 1.2303), respectively,
based on 5000 samples.

2.9 Analysis of time-to-event endpoints

This section focuses on the colorectal cancer trial from Case study 4. First, we
conduct the analysis of the primary time-to-event endpoint in the trial (time to
death) using log-rank scores and adjusting for the number of days from initial cancer
diagnosis to start of study drug administration and the sum of the longest lesion
diameters in millimeters at baseline. We perform a stratified analysis of overall
survival based upon the KRAS gene, weighting each center using Mantel-Haenszel
weights. Further, we combine treatment estimates across strata prior to covariance
adjustment (Program 2.14). Note that for time-to-event outcomes, the time until the
event occurs or when the patient is censored (os time) is provided in EXPOSURES. The
binary outcome of whether a patient experienced an event or not (here, os event,
where 1 implies an event occurred, and 0 implies the patient was censored) is
provided in OUTCOMES.

PROGRAM 2.14 Overall Survival Using Log-rank Scores with Adjustment for Covariates

%NParCov4(OUTCOMES = os_event, COVARS = days length, EXPOSURES = os_time,

STRATA = mutant, HYPOTH = NULL, TRTGRPS = treatment, TRANSFORM =

LOGRANK, COMBINE = FIRST, EXACT = YES, NREPS = 5000, SEED = 45,

DSNIN = PROSTATE, DSNOUT = OS);

The criterion for covariate imbalance Q = 1.19 is distributed χ2
(2). The asymptotic

p-value for the criterion is 0.5514, and the exact p-value based on 5000 permutation
samples is 0.5484. It appears that randomization provided reasonably equal distri-
butions of covariates between the two treatments. The treatment estimate, which
represents the difference in average log-rank scores for overall survival, is presented
in Table 2.13.

TABLE 2.13 Overall Survival Using Log-rank Scores with Adjustment for Covariates

Outcome Treatment Effect Standard Error Test Statistic P-value

Overall Survival 0.0299 0.1749 0.0292 0.8642

The exact p-value based on 5000 permutation samples is 0.8712. The positive
treatment estimate implies that the treatment has decreased overall survival from
colorectal cancer, relative to control. However, this result is not significant so we
conclude that there is no difference between the treatments when adjusting for the
covariates. A similar analysis for the other time-to-event endpoint in the colorectal
cancer trial (time to disease progression) is performed in Program 2.15.

PROGRAM 2.15 Progression-Free Survival Using Log-rank Scores with Adjustment for Covariates

%NParCov4(OUTCOMES = pfs_event, COVARS = days length, EXPOSURES = pfs_time,

STRATA = mutant, HYPOTH = NULL, TRTGRPS = treatment, TRANSFORM =

LOGRANK, COMBINE = FIRST, EXACT = YES, NREPS = 5000, SEED = 45,

DSNIN = PROSTATE, DSNOUT = PFS);

In this case, the criterion for covariate imbalance Q = 1.19, which is distributed
χ2

(2). The asymptotic p-value for the criterion is 0.5514, and the exact p-value based
on 5000 permutation samples is 0.562. It appears that randomization provided
reasonably equal distributions of covariates between the two treatments. The
treatment estimate, which represents the difference in average log-rank scores for
progression-free survival, is presented in Table 2.14.

TABLE 2.14 Progression-Free Survival Using Log-rank Scores with Adjustment for Covariates

Outcome Treatment Effect Standard Error Test Statistic P-value

PFS 0.0090 0.1750 0.0026 0.9592

The exact p-value based on 5000 permutation samples is 0.9600. The positive
treatment estimate implies that the treatment has decreased progression-free survival
from colorectal cancer, relative to control. However, this result is not significant, so
we conclude that there is no difference between the treatments when adjusting for
the covariates.

A joint test of the overall and progression-free survival can be performed with
multiple calls to %NParCov4 to test the hypothesis

H0 : fos = 0, fpfs = 0,

where fos and fpfs are the treatment effects for overall and progression-free survival,
respectively. The OUTDAT SURV data sets from the two runs above provide the
log-rank scores (LOGRANK OS EVENT or LOGRANK PFS EVENT). These data sets can be
merged together, and a joint analysis of overall and progression-free survival can be
performed using Program 2.16.

PROGRAM 2.16 Joint Analysis of Overall and Progression-Free Survival Using Log-rank Scores
with Adjustment for Covariates

%NParCov4(OUTCOMES = logrank_os_event logrank_pfs_event, COVARS = days length,

EXPOSURES =, STRATA = mutant, HYPOTH = NULL, TRTGRPS = treatment,

TRANSFORM = NONE, COMBINE = FIRST, EXACT = YES, NREPS = 5000, SEED = 45,

DSNIN = PROSTATE);

Notice that EXPOSURES is left unspecified since these values have already been
accounted for in the LOGRANK OS EVENT and LOGRANK PFS EVENT outcomes. Further,
no transformation is needed, so TRANSFORM = NONE. Defining C = I2, an identity
matrix of order two, and using similar methods and code as those for Program 2.6,
the two degree-of-freedom bivariate test is computed in Program 2.17.

PROGRAM 2.17 Bivariate Test of Overall and Progression-Free Survival

proc iml;

use _OUTDAT_DEPTEST;

read all var{beta} into B;

close;

use _OUTDAT_COVBETA;

read all var{logrank_os_event logrank_pfs_event} into V;

close;

C = I(2);

Q = t(B) * t(C) * inv(C * V * t(C)) * C * B;

p = 1 - probchi(Q, nrow(C));

print Q p;

quit;

Here, QBivariate = 0.0295 with asymptotic p = 0.9854. It is natural to ask if an
exact p-value is available for this example. As described in Appendix B, the exact
methodologies of %NParCov4 are limited to a single outcome variable. However, it
is possible to write additional code to repeatedly call %NParCov4. Though repeated
calls to %NParCov4 might not be the most efficient procedure programmatically,
it does enable the analyst to compute exact results for more general problems.

In Program 2.18, the QBivariate statistic is computed for each permutation sample,
and the proportion of these QBivariate statistics that is greater than or equal to the
QBivariate statistic computed from the observed sample constitutes the p-value.

PROGRAM 2.18 Exact Bivariate Test of Overall and Progression-Free Survival

%let nreps = 5000;

%let seed = 50;

options nonotes;

proc sort data = both;

by treatment patient;

run;

proc multtest permutation nocenter noprint seed=&seed nsample=&nreps

outsamp=samp_null1(drop=_obs_) data=both;

class treatment;

strata mutant;

test mean(logrank_os_event logrank_pfs_event days length);

run;

data samp_null2;

set samp_null1;

*** convert treatment back to numeric ***;

treatment =_class_+0;

*** convert strata back to numeric ***;

mutant =_stratum_+0;

drop _class_;

run;

*** sample 0 is the observed sample ***;

data samp_null3;

set both(in=a) samp_null2;

if a then _sample_=0;

run;

%macro BIVARIATE;

%do rep = 0 %to &nreps;

data temp;

set samp_null3;

where _sample_ = &rep;

run;

%put PROGRESS = &rep;

%NPARCOV4(outcomes = logrank_os_event logrank_pfs_event,

covars = days length,

trtgrps = treatment,

strata = mutant,

hypoth = NULL,

transform = NONE,

combine = FIRST,

c = 1,

dsnin = temp,

dsnout = outdat,

exact = NO,

details = NO);

%if &rep = 0 %then %do;

data allbetas;

set _outdat_deptest(keep = outcomes beta);

_sample_ = &rep;

run;

data allcovs;

set _outdat_covbeta;

_sample_ = &rep;

run;

%end;

%else %do;

data allbetas;

set allbetas _outdat_deptest(keep = outcomes beta in = inb);

if inb then _sample_ = &rep;

run;

data allcovs;

set allcovs _outdat_covbeta(in = inb);

if inb then _sample_ = &rep;

run;

%end;

proc datasets nolist;

delete temp;

quit;

%end;

%mend BIVARIATE;

%BIVARIATE;

proc iml;

use allbetas;

read all var{beta} into B;

close;

use allcovs;

read all var{logrank_os_event logrank_pfs_event} into V;

close;

C = I(2);

Q0 = t(B[1:2]) * C * inv(C * V[1:2,] * t(C)) * C * B[1:2];

twosided = 0;

do i = 2 to &nreps+1;

Btemp = B[1+(i-1)*ncol(V):i*ncol(V)];

Vtemp = V[1+(i-1)*ncol(V):i*ncol(V),];

Q = t(Btemp) * t(C) * inv(C * Vtemp * t(C)) * C * Btemp;

twosided = twosided + (Q >= Q0);

end;

twosided = twosided / &nreps;

print Q0 twosided;

quit;

options notes;

Program 2.18 generates 5000 permutation samples to compute an essentially
exact p-value = 0.9836. DETAILS = NO and OPTIONS NONOTES are used to minimize
the amount of information saved to the SAS log while these simulations are running.
Note that in practice, an exact p-value is typically not useful when the asymptotic
p-value exceeds 0.15. Computing an exact p-value here was for illustration more
than anything else.

This entire analysis could have been conducted using Wilcoxon scores. An
alternative analysis that might be of interest is to conduct a joint analysis of an
endpoint using both Wilcoxon and log-rank scores. For example, in the above
description, merely replace the progression-free survival call with an additional
analysis for overall survival using Wilcoxon scores (TRANSFORM = WILCOXON) to test
the null hypothesis

H0 : fWilcoxon = 0, flogrank = 0,

where fWilcoxon and flogrank are the treatment effects for overall survival. They
are computed using the Wilcoxon or log-rank scores, respectively. What is the
rationale for such an analysis? In some instances, it may not be clear which test
might be most appropriate for a given data set. For example, the log-rank test
assumes proportional hazards between the two treatments that might not hold in
practice. Wilcoxon scores place greater emphasis on earlier event times (Collett,
2015). If the proportional hazards assumption holds, it might be unclear which test
might be more appropriate. In oncology, the log-rank test is often more powerful to
assess long-term survival. However, in analgesic studies where interest is in earlier
pain relief, the Wilcoxon test might be preferred. See Tangen and Koch (1999b) for
an illustration of this particular bivariate test.

2.10 Summary

This chapter presented the nonparametric, randomization-based analysis of covari-
ance methodology and its application to the analysis of clinical endpoints. The
methodology was implemented using the %NParCov4 macro that contains functional-
ity for many of the analysis methods described to date, with the ability to compute
essentially-exact p-values and confidence intervals. It is recommended that advanced
randomization-based methods be specified as the primary analysis because of their
minimal assumptions and the availability of essentially-exact p-values. Model-based
methods, such as a logistic regression model in the case of binary outcomes, can be
used in a supportive fashion to assess the effects of covariates, potential interactions
of treatment with covariates, and the generalizability of findings to populations not
represented by the distributions of subjects in a clinical trial.

Appendix

A Theoretical Details

A.1 Non-Parametric Randomization-Based Analysis of Covariance

Suppose there are two randomized treatment groups, and there is interest in a
nonparametric, randomization-based comparison of r outcomes adjusting for t
covariates within a single stratum. Let ni be the sample size from a population of
Ni >> ni patients, ȳi be the r-vector of outcome means, and x̄i be the t-vector of
covariate means of the ith treatment group, i = 1, 2. Let

fi =

[
g(ȳi)
x̄i

]
,

where g(·) is a twice-differentiable function. Further, define cov(fi) = Vfi = DiViDi,
where

Vi = cov

([
ȳi
x̄i

])
and Di is a diagonal matrix with the first derivatives g′(ȳi) corresponding to the
first r cells of the diagonal with ones for the remaining t cells of the diagonal.

Define f = f2 − f1 and Vf = Vf1 + Vf2 and let Ir be an identity matrix of
dimension r and 0(t×r) be a t× r matrix of zeros and fit the model

E[f ]=̂

[
Ir

0(t×r)

]
β̂ = Xβ̂

using weighted least squares with weights based on V −1
f , with =̂ meaning ”is

estimated by”. β is an r × 1 vector of covariate-adjusted treatment effects. The
weighted least squares estimator is



Chapter 2 Advanced Randomization-based Methods 89

β̂ = (X>V −1
f X)−1X>V −1

f f

and a consistent estimator for cov(β̂) is (X>V −1
f X)−1. There are two ways to

estimate Vf . Under the null hypothesis of no treatment difference,

Vi =
1

ni(n1 + n2 − 1)

{
2∑
i=1

ni∑
l=1

[
(yil − ȳ)(yil − ȳ)> (yil − ȳ)(xil − x̄)>

(xil − x̄)(yil − ȳ)> (xil − x̄)(xil − x̄)>

]}
,

where ȳ and x̄ are the means for the outcomes and covariates across both treatment
groups, respectively. Under the alternative,

Vi =
1

ni(ni − 1)

{
ni∑
l=1

[
(yil − ȳi)(yil − ȳi)> (yil − ȳi)(xil − x̄i)>
(xil − x̄i)(yil − ȳi)> (xil − x̄i)(xil − x̄i)>

]}
.

In general, the variance under the null would be the principal structure for
producing p-values for hypothesis testing under the null when the null corresponded
to no difference between treatments (or more formally, the strong null that each
patient would have the same outcome, regardless of the assigned treatment). The
variance under the alternative will be used for computing confidence intervals.
Here, the standard errors address replicated random samples from a population for
which the randomized population is arguably a random sample. P -values under the
alternative might be useful since they will agree with the confidence interval, at least
when both are based on normal approximations. Throughout this chapter, we take
the approach that p-values will be computed under the null, with corresponding
confidence intervals computed under the alternative hypothesis.

A test to compare the two treatments for outcome yj , j = 1, 2, . . . , r, with
adjustment for covariates is

Qj =
β̂2
j

Vβ̂j
.

With sufficient sample size, Qj is approximately distributed χ2
(1). A criterion for

the random imbalance of the covariates between the two treatment groups is

Q = (f −Xβ̂)>V −1
f (f −Xβ̂)

which is approximately distributed χ2
(t). This criterion should not be considered

a test for validity of nonparametric, randomization-based analysis of covariance,
but merely as a measure of the degree of imbalance among covariates between the
treatments. Due to the randomization requirement of the methodology, significant
imbalance detected among the covariates should be interpreted as a chance finding.
Adjusting for the covariates in the analysis removes this imbalance. The results of an
unadjusted analysis would be confounded with the imbalance among the covariates.
Small p-values for the covariate imbalance criterion are more appropriately referred
to as ”atypical relative to chance” rather than ”significant.”

Under the alternative hypothesis, asymptotic confidence intervals for β̂j can be

defined as β̂j ± Z(1−α/2)V
1/2

β̂j
, where Z(1−α/2) is a quantile from a standard normal

random variable.
In many situations, a linear transformation of ȳi is appropriate, e.g., g(ȳi) = ȳi.

Here, Di will be an identity matrix of dimension (r × t). In %NParCov4, the above
methods correspond to TRANSFORM = NONE and COMBINE = NONE.

A.2 Logistic Transformation

If ȳi consists of (0,1) binary outcomes, a logistic transformation of the elements of
ȳi might be more appropriate. Let g(ȳi) = logit(ȳi) where logit(·) transforms each

element ȳij of ȳi to loge(ȳij/(1− ȳij)). The first r elements of f correspond to

loge
ȳ2j(1− ȳ1j)

ȳ1j(1− ȳ2j)
,

which is the loge-odds ratio of the jth outcome. Under the null hypothesis, the first r
elements of the diagonal of Di are {ȳj(1− ȳj)}−1, where ȳj is the mean for outcome
yj across both treatment groups. Under the alternative, the first r elements of the

diagonal of Di are {ȳij(1− ȳij)}−1. The β̂j and associated confidence intervals can
be exponentiated to obtain estimates and confidence intervals for the odds ratios.
These methods correspond to TRANSFORM = LOGISTIC and COMBINE = NONE.

A single ordinal outcome with (r+ 1) levels can be analyzed as r binary outcomes
(cumulative across levels) using a proportional odds model. Logistic transformations
are applied to the r outcomes as described above. A test of the homogeneity of the
r odds ratios would be computed as

Qc = β̂>C>(CVβ̂C
>)−1Cβ̂,

where C = [ I(r−1) −1(r−1) ], I(r−1) is an identity matrix of dimension (r − 1)
and −1(r−1) is a vector of length (r − 1) containing elements −1. With sufficient
sample size, Qc is approximately distributed χ2

(r−1).
If homogeneity applies, you can use the simplified model

E[f ] =

[
1r
0t

]
β̂R = XRβ̂R.

For ordinal outcomes, β̂R would be the common odds ratio across cumulative
outcomes in a proportional odds model. The criterion for random imbalance for
this simplified model will be a joint test of random imbalance and proportional
odds, which is approximately χ2

(t+r−1). These methods correspond to TRANSFORM =

PODDS and COMBINE = NONE. %NParCov4 computes the test of homogeneity and the
reduced model when PODDS is selected. Should the p-value for Qc be such that it
is ≤ α, the proportional odds assumption is violated, which implies that the full
model is more appropriate (obtained with TRANSFORM = LOGISTIC).

A.3 Log-Ratio of Two Outcome Vectors

Suppose there was interest in the analyses of the ratio of means while adjusting for
the effect of covariates. Let g(ȳi) = loge(ȳi) where loge(·) transforms each element
ȳij of ȳi to loge(ȳij). The first r elements of f correspond to

loge(ȳ2j)− loge(ȳ1j) = loge

(
ȳ2j

ȳ1j

)
,

which is the loge-ratio of the jth outcome. Under the null hypothesis, the first r
elements of the diagonal of Di are ȳ−1

j , where ȳj is the mean for outcome yj across
both treatment groups. Under the alternative, the first r elements of the diagonal
of Di are ȳ−1

ij . The β̂j and associated confidence intervals can be exponentiated to
obtain estimates and confidence intervals for the ratios of means. These methods
correspond to TRANSFORM = LOGRATIO and COMBINE = NONE.

A.4 Log-Incidence Density Ratios

Suppose each of the r outcomes has an associated exposure so that ēi is an r-vector
of average exposures. Let

fi =

 loge(ȳi)
loge(ēi)
x̄i

 ,

where loge(·) transforms each element ȳij of ȳi to loge(ȳij) and each element ēij of
ēi to loge(ēij) and let

M0 =

[
Ir −Ir 0(r×t)

0(t×r) 0(t×r) It

]
,

where 0(r×t) is an r × t matrix of zeros and Ir and It are identity matrices of
dimension r and t, respectively. Define

gi = M0fi =

[
loge(ȳi)− loge(ēi)

x̄i

]
so that cov(gi) = Vgi = MiViM

>
i , where

Vi = cov

 ȳi
ēi
x̄i


and

Mi =

[
Dȳi −Dēi 0(r×t)

0(t×r) 0(t×r) It

]
.

The first r elements of gi are loge-incidence densities so that if g = g2 − g1, the first
r elements of g correspond to

loge

(
ȳ2j

ē2j

)
− loge

(
ȳ1j

ē1j

)
= loge


(
ȳ2j
ē2j

)
(
ȳ1j
ē1j

)
 ,

which is the loge-incidence density ratio of the jth outcome. Under the null hypoth-
esis, Dȳi is a diagonal matrix with diagonal elements ȳ−1

j , and Dēi is a diagonal

matrix with diagonal elements ē−1
j , where ȳj and ēj are the means for outcome yj

and exposure ej across both treatment groups, respectively. Under the alternative
hypothesis, Dȳi is a diagonal matrix with diagonal elements ȳ−1

ij . And Dēi is a

diagonal matrix with diagonal elements ē−1
ij . The β̂j and associated confidence

intervals can be exponentiated to obtain estimates and confidence intervals for the
incidence density ratios. These methods correspond to TRANSFORM = INCDENS and
COMBINE = NONE.

A.5 Wilcoxon and Log-rank Scores for Time-to-Event Outcomes

Unlike Sections A.2, A.3, and A.4 where a transformation is applied to the means of
one or more outcomes, %NParCov4 can transform subject-level event flags and times
into Wilcoxon or log-rank scores to perform analyses of time-to-event outcomes.
Scores are defined from binary outcome flags (OUTCOMES) that define whether an
event occurs (= 1) or not (= 0) and the exposure times (EXPOSURES) when the event
occurs or the last available time at which an individual is censored (e.g., when the
subject does not experience the event). The definitions are from the appendix of
Tangen and Koch (1999b). The Wilcoxon scores are closely related to the Prentice
generalized Wilcoxon scores presented in Kalbfleisch and Prentice (1980).

Suppose there are L unique failure times such that y(1) < y(2) < . . . < y(L).
Assume that gζ is the number of failures that occur at yζ and that there are
yζ1 , yζ2 , . . . , yζνζ censored values in [y(ζ), y(ζ+1)), where νζ is the number of censored

subjects in the interval and ζ = 1, 2, . . . , L. Further, assume y(0) = 0, y(L+1) =∞,
and g0 = 0.

Log-rank scores are expressed as

cζ = 1−
ζ∑

ζ′=1

gζ′

Nζ′
and Cζ = −

ζ∑
ζ′=1

gζ′

Nζ′

and C0 = 0, where cζ and Cζ are the scores for the uncensored and censored
subjects, respectively, in the interval [y(ζ), y(ζ+1)) for ζ = 1, 2, . . . , L where Nζ =

n−
∑ζ−1
ζ′=0(νζ′ + gζ′) is the number of subjects at risk at t(ζ) − 0 and n = n1 + n2

is the sample size. To conduct an unstratified analysis using log-rank scores, set
TRANSFORM = LOGRANK and COMBINE = NONE.

Wilcoxon scores are expressed as

cζ = 2

ζ∏
ζ′=1

(
Nζ′ − gζ′
Nζ′

)
− 1 and Cζ =

ζ∏
ζ′=1

(
Nζ′ − gζ′
Nζ′

)
− 1

and C0 = 0, where cζ and Cζ are the scores for the uncensored and censored subjects,
respectively, in the interval [y(ζ), y(ζ+1)) for ζ = 1, 2, . . . , L where Nζ is the number of
subjects at risk at t(ζ)−0. To conduct an unstratified analysis using Wilcoxon scores,
set TRANSFORM = WILCOXON and COMBINE = NONE. %NParCov4 computes survival
scores, summarizes as ȳi and analyzes as in Section A.1 using a linear transformation.

A.6 Stratification

Stratification can be handled in one of three ways:

1. Estimate β̂h for each stratum h, h = 1, 2, . . . ,H, and combine across strata to
obtain

β̄ =

∑
h whβ̂h∑
h wh

,

where

wh =

(
nh1nh2

nh1 + nh2

)c
and 0 ≤ c ≤ 1, with c = 1 for Mantel-Haenszel weights and c = 0 for equal strata
weights. A criterion for random imbalance is performed within each stratum and
then summed to give an overall test of imbalance across all strata. This method
is more appropriate for large sample size per stratum (nhi ≥ 30 and preferably
nhi ≥ 60). This corresponds to COMBINE = LAST.

2. Form

fw =

∑
h whfh∑
h wh

first and fit a model for covariance adjustment to determine β̂w. This method is
more appropriate for small to moderate sample size per stratum. This corresponds
to COMBINE = FIRST.

3. For transformations in Sections A.2, A.3, and A.4, define[
ȳwi
x̄wi

]
=

(∑
h

wh

)−1

×
∑
h

(
wh

[
ȳhi
x̄hi

])
first, then proceed with the transformation and finally apply covariance adjust-
ment. This method is more appropriate for small sample size per stratum and
corresponds to COMBINE = PRETRANSFORM.

A.7 Resampling-Based Exact Methods

%NParCov4 contains options for computing essentially exact p-values under HYPOTH
= NULL and confidence intervals under HYPOTH = ALT for a single outcome. Under
EXACT = YES and HYPOTH = NULL, %NParCov4 uses PROC MULTTEST with the
PERMUTATION option to compute NREPS data sets. In this case, the sampling of
records to either treatment (within strata) is performed without replacement. Under
EXACT = YES and HYPOTH = ALT, %NParCov4 uses PROC MULTTEST with the
BOOTSTRAP option and a BY statement for treatment to compute NREPS data sets
where the sampling of records occurs with replacement by treatment (within strata).
Further, under EXACT = YES and HYPOTH = ALT, n jackknife data sets are created
where a single observation is deleted from each jackknife data set within strata,
where n is the total number of observations in the analysis. The jackknife data sets
are used in the computation of acceleration for the bias-corrected and accelerated
(BCa) intervals.

Exact p-values are obtained for EXACT = YES and HYPOTH = NULL. An exact
criterion for the random imbalance of covariates between the two treatment groups
computes Qm, m = 1, 2, . . . , B, for each of B resampled data sets. The p-value is
calculated as #(Qm ≥ Q0)/B, where Q0 is the criterion for random imbalance for

the observed data set. Under the null hypothesis, the treatment parameter β̂m,
m = 1, 2, . . . , B, is computed for each resampled data set and is compared to the
treatment parameter of the observed data set (β̂0) to calculate p-values:

1. two-sided: #(|β̂m| ≥ |β̂0|)/B,

2. one-sided lower: #(β̂m ≤ β̂0)/B,

3. one-sided upper: #(β̂m ≥ β̂0)/B.

Under EXACT = YES and HYPOTH = ALT, bias-corrected and accelerated intervals
(BCa) are computed using the B bootstrap samples and the n jackknife samples
(Efron and Tibshirani, 1993). The 100(1 − α)% BCa interval is the α1 and α2

percentiles of all B bootstrap estimates and is defined as BCa = (β̂(α1), β̂(α2)), where

α1 = Φ

(
b̂+

b̂+ Φ−1(α/2)

1− â(b̂+ Φ−1(α/2))

)
, α2 = Φ

(
b̂+

b̂+ Φ−1(1− α/2)

1− â(b̂+ Φ−1(1− α/2))

)
,

b̂ = Φ−1
(

#(β̂m<β̂0)
B

)
is the bias-correction,

â =

∑H
h=1 n

−3
h

∑nh
i=1(β̂∗hi − β̄∗h)3

6
(∑H

h=1 n
−2
h

∑nh
i=1(β̂∗hi − β̄∗h)2

)3/2

is the acceleration (Davis and Hinkley 1997); β̂∗hi is the estimate from the ith
jackknife sample in the hth strata; β̄∗h is the mean of the nh jackknife estimates
from the hth strata, Φ(·) is the CDF and Φ−1(·) is the inverse-CDF of the standard
normal distribution, and α is the type I error.

100(1 − α)% percentile confidence intervals are also calculated where (β̂(α/2),

β̂(1−α/2)) are the α/2 and 1− α/2 percentiles of the B estimates. When COMBINE

= PRETRANSFORM or COMBINE = FIRST, acceleration is computed as if in a single
stratum using β̂w. An exact two-sided p-value criterion for covariate imbalance
(COVXACT) is calculated as #(|Qm| ≥ |Q0|)/B. While exact analyses are limited to a
single outcome in %NParCov4, Program 2.18 in Section 2.9 illustrates how the macro
can be repeatedly called to generate exact results for more general problems.

B SAS Macro %NParCov4

B.1 General Use and Options

%NParCov4 can be accessed by your program by including the lines:

filename nparcov4 "&path";

%include nparcov4(NParCov4) / nosource2;

or

%include "&path./nparcov4.sas".

Data must be set up so there is one line per experimental unit, and there cannot be
any missing data. Records with missing data should be deleted or made complete
with imputed values prior to analysis. The program does not automatically write
output to the SAS listing. The results of individual data sets can be displayed
using PROC PRINT. PROC DATASETS will list available data sets generated by
%NParCov4 at the end of the SAS log. %NParCov4 has the following options:

1. OUTCOMES: List of outcome variables for the analysis. All outcome variables are
numeric and at least one outcome is required. Binary or time-to-event outcomes
that use TRANSFORM = LOGISTIC, PODDS, LOGRANK, or WILCOXON can only have
(0,1) values. Here, 1 should indicate that an event occurs, while 0 indicates that
an event does not occur (or, in the case of a survival endpoint, that an individual
is censored).

2. COVARS: List of covariates for the analysis. All covariates must be numeric;
categorical variables must be re-expressed using one or more indicator variables.
COVARS can be left unspecified (blank) for stratified, unadjusted results.

3. EXPOSURES: Required numeric exposures for TRANSFORM = LOGRANK, WILCOXON,
or INCDENS. The same number of exposure variables as OUTCOMES is required,
and the order of the variables in EXPOSURES should correspond to the variables
in OUTCOMES.

4. TRTGRPS: This is a required, single two-level numeric variable that defines the
two randomized treatments or groups of interest. Differences are based on higher
number (TRT2) - lower number (TRT1). For example, if the variable is coded
(1,2), differences computed are treatment 2 - treatment 1.

5. STRATA: A single numeric variable that defines strata, this variable can represent
strata for cross-classified factors. Specify NONE (default) for one stratum (i.e.,
no stratification). Note that each stratum must have at least one observation
per treatment when HYPOTH = NULL and at least two observations per treatment
when HYPOTH = ALT.

6. C: Used in calculation of strata weights for all methods. 0 ≤ C ≤ 1 (default = 1).
A 0 assigns equal weights to strata, while 1 is for Mantel-Haenszel weights.

7. COMBINE: Defines how the analysis will accommodate STRATA. Choose one.

(a) NONE (default) performs analysis as if data are from a single stratum.

(b) LAST performs covariance adjustment within each stratum, and then takes
weighted averages of parameter estimates with weights based on stratum
sample size (appropriate for larger strata).

(c) FIRST obtains a weighted average of treatment group differences (post-
transformation, if applicable) across strata, then performs covariance adjust-
ment (appropriate for smaller strata).

(d) PRETRANSFORM (available for TRANSFORM = LOGISTIC, PODDS, LOGRATIO,
or INCDENS) obtains a weighted average of treatment group means (pre-
transformation) across strata, performs a transformation, and then applies
covariance adjustment (appropriate for very small strata).

8. HYPOTH: Determines how variance matrices are computed.

(a) NULL applies the assumption that the means and covariance matrices of
the treatment groups are equal and therefore computes a single covariance
matrix for each stratum.

(b) ALT computes a separate covariance matrix for each treatment group within
each stratum.

9. TRANSFORM: Applies transformations to outcome variables. Choose one.

(a) NONE (default) applies no transformation and analyzes means for one or more
outcome variables as provided.

(b) LOGISTIC applies the logistic transformation to the means of one or more
(0,1) outcomes.

(c) PODDS applies the logistic transformation to multiple means of (0,1) outcomes
that together form a single ordinal outcome.

(d) LOGRATIO applies the log-transformation based on the natural logarithm to
one or more ratios of outcome means.

(e) INCDENS uses number of events experienced in OUTCOMES and EXPOSURES

to calculate and analyze the log-transformation of one or more ratios of
incidence density.

(f) LOGRANK uses event flags in OUTCOMES (1 = event, 0 = censored) and
EXPOSURES to calculate and analyze log-rank scores for one or more time-to-
event outcomes.

(g) WILCOXON uses event flags in OUTCOMES (1 = event, 0 = censored) and
EXPOSURES to calculate and analyze Wilcoxon scores for one or more time-
to-event outcomes.

10. ALPHA: Confidence limit for intervals (default = 0.05).

11. EXACT: Provides essentially exact analyses for a single outcome. Choose one.

(a) NO (default) does not perform resampling analyses;

(b) YES performs additional resampling-based analysis. If HYPOTH = ALT, boot-
strap resampling is performed with observations sampled with replacement
within treatment. If HYPOTH = NULL, permutation resampling is performed
with observations sampled without replacement across treatments. This is
equivalent to shuffling the treatment codes of the observations.

12. NREPS: Number of random data sets to generate when EXACT = YES (default =
1000).

13. SEED: Random seed when EXACT = YES (default = 0).

14. DSNIN: Input data set. Required.

15. DSNOUT: Prefix for output data sets. Required. For example, if DSNOUT is set to
OUTDAT, the following data sets will be created:

(a) OUTDAT COVTEST contains the criterion for covariate imbalance. When EXACT

= YES and HYPOTH = NULL, the exact criterion for covariate imbalance
(COVXACT) is calculated.

(b) OUTDAT DEPTEST contains tests, treatment estimates, and ratios (for
TRANSFORM = LOGISTIC, PODDS, LOGRATIO, or INCDENS) for outcomes.

(c) OUTDAT CI contains confidence intervals for outcomes when HYPOTH = ALT.

(d) OUTDAT RATIOCI contains confidence intervals for odds ratios (TRANSFORM =

PODDS or LOGISTIC), ratios (TRANSFORM = LOGRATIO), or incidence densities
(TRANSFORM = INCDENS) when HYPOTH = ALT.

(e) OUTDAT EXACT provided when EXACT = YES

i. When HYPOTH = NULL, the data set contains essentially exact, one- and
two-sided p-values for the outcome.

ii. When HYPOTH = ALT, the data set contains percentile, bias-corrected,
and accelerated confidence intervals. For TRANSFORM = LOGISTIC,
PODDS, LOGRATIO, or INCDENS, the exact percentile, bias-corrected, and
accelerated confidence intervals for the ratios are also provided.

(f) OUTDAT HOMOGEN contains the test for homogeneity of odds ratios for
TRANSFORM = PODDS.

(g) OUTDAT BETASAMP contains the resampled values of the treatment param-
eter (BETASAMP). The ratios for TRANSFORM = LOGISTIC, PODDS, LOGRATIO,
or INCDENS, are also provided (EXPBETASAMP). The observed treatment pa-
rameter is provided (labeled OBSERVED) and for either of the following:

i. HYPOTH = NULL, parameters from the permutation data sets are labeled
PERMUTATION.

ii. HYPOTH = ALT, parameters from bootstrap data sets are labeled
BOOTSTRAP, while parameters from jackknife data sets are labeled
JACKKNIFE.

(h) OUTDAT SURV contains survival scores for TRANSFORM = LOGRANK or
WILCOXON. The scores from this data set from multiple calls of %NParCov4

can be used with TRANSFORM = NONE to analyze log-rank and Wilcoxon
scores simultaneously or to analyze survival outcomes with non-survival
outcomes.

(i) OUTDAT COVBETA contains the covariance matrix for β̂. The estimates from
data set OUTDAT DEPTEST and the covariance matrix can be used to test
hypotheses of linear combinations of the β̂.

(j) OUTDAT STRATABETA contains the strata-specific estimates for β̂ when
COMBINE = LAST. These estimates can be used to compute acceleration for
BCa intervals (Appendix A.7) when repeatedly calling %NParCov4 for more
general problems.

16. SYMSIZE: Define space requirements for SAS/IML. Should be increased for larger
data sets or large NREPS (default = 20000).

17. DETAILS: Print basic analysis information and the PROC DATASETS of analysis
results to the SAS log (default = YES).

B.2 Content of Output Data Sets

Output data sets contain the following variables:

1. OUTDAT COVTEST

(a) TYPE = COVTEST.

(b) TRT1 is the value for control (lower value numerically).

(c) TRT2 is the value for treatment (higher value numerically).

(d) STRATA contains the strata variable.

(e) COVARIATES contains the list of covariates.

(f) OUTCOMES contains the list of outcomes.

(g) EXPOSURES contains the list of exposures for time-to-event endpoints.

(h) HYPOTHESIS is either NULL or ALT.

(i) H is the number of strata levels.

(j) DF is the degrees of freedom.

(k) Q is the test statistic.

(l) PVALUE is the p-value.

(m) COVXACT is the exact p-value when EXACT = YES and HYPOTH = NULL.

2. OUTDAT DEPTEST.

(a) TYPE = DEPTEST.

(b) TRANSFORM is the transformation applied to the outcomes.

(c) TRT1 is the value for control (lower value numerically).

(d) TRT2 is the value for treatment (higher value numerically).

(e) STRATA contains the strata variable.

(f) COVARIATES contains the list of covariates.

(g) OUTCOMES contains the list of outcomes.

(h) EXPOSURES contains the list of exposures for time-to-event endpoints.

(i) HYPOTHESIS is either NULL or ALT.

(j) H is the number of strata levels.

(k) DF is the degrees of freedom.

(l) BETA is the treatment estimate.

(m) SEBETA is the standard error of the treatment estimate.

(n) RATIO is the exponentiated treatment estimate for TRANSFORM = LOGISTIC,
PODDS, LOGRATIO, INCDENS.

(o) Q J is the test statistic.

(p) PVALUE is the p-value.

3. OUTDAT CI.

(a) TYPE = CI.

(b) TRT1 is the value for control (lower value numerically).

(c) TRT2 is the value for treatment (higher value numerically).

(d) STRATA contains the strata variable.

(e) COVARIATES contains the list of covariates.

(f) OUTCOMES contains the list of outcomes.

(g) EXPOSURES contains the list of exposures for time-to-event endpoints.

(h) HYPOTHESIS is either NULL or ALT.

(i) H is the number of strata levels.

(j) BETA is the treatment estimate.

(k) SEBETA is the standard error of the treatment estimate

(l) LOWER is the lower limit of the 100(1 − α)% confidence interval for the
treatment estimate.

(m) UPPER is the upper limit of the 100(1 − α)% confidence interval for the
treatment estimate.

(n) ALPHA is the value of α.

4. OUTDAT RATIOCI

(a) TYPE = RATIOCI.

(b) TRT1 is the value for control (lower value numerically).

(c) TRT2 is the value for treatment (higher value numerically).

(d) STRATA contains the strata variable.

(e) COVARIATES contains the list of covariates.

(f) OUTCOMES contains the list of outcomes.

(g) EXPOSURES contains the list of exposures for time-to-event endpoints.

(h) HYPOTHESIS is either NULL or ALT.

(i) H is the number of strata levels.

(j) RATIO is the exponentiated treatment estimate.

(k) RATIO LOWER is the lower limit of the 100(1 − α)% confidence interval for
the exponentiated treatment estimate.

(l) RATIO UPPER is the upper limit of the 100(1− α)% confidence interval for
the exponentiated treatment estimate.

(m) ALPHA is the value of α.

5. OUTDAT EXACT.

(a) TYPE = EXACT.

(b) TRT1 is the value for control (lower value numerically).

(c) TRT2 is the value for treatment (higher value numerically).

(d) STRATA contains the strata variable.

(e) COVARIATES contains the list of covariates.

(f) OUTCOMES contains the list of outcomes.

(g) HYPOTHESIS is either NULL or ALT.

(h) H is the number of strata levels.

(i) TWOSIDED is the two-sided, exact p-value to test the treatment effect.

(j) ONE LOWER is the lower one-sided, exact p-value to test the treatment effect.

(k) ONE UPPER is the upper one-sided exact p-value to test the treatment effect.

(l) ALPHA is the value of α.

(m) NREPS is the number of resamples.

(n) SEED is the value of seed.

(o) BIAS is b̂, the bias used in the calculation of the BCa interval.

(p) ACCEL is â, the acceleration used in the calculation of the BCa interval.

(q) ALPHA LOW is α1 used in the calculation of the BCa interval.

(r) ALPHA HI is α2 used in the calculation of the BCa interval.

(s) BCA LOWER is the lower limit of the BCa interval for the treatment effect.

(t) BCA UPPER is the upper limit of the BCa interval for the treatment effect.

(u) PCT LOWER is the lower limit of the percentile interval for the treatment
effect.

(v) PCT UPPER is the upper limit of the percentile interval for the treatment
effect.

(w) BCA RATIO LOWER is the lower limit of the BCa interval for the exponentiated
treatment effect for TRANSFORM = LOGISTIC, PODDS, LOGRATIO, INCDENS.

(x) BCA RATIO UPPER is the upper limit of the BCa interval for the exponentiated
treatment effect for TRANSFORM = LOGISTIC, PODDS, LOGRATIO, INCDENS.

(y) PCT RATIO LOWER is the lower limit of the percentile interval for the expo-
nentiated treatment effect for TRANSFORM = LOGISTIC, PODDS, LOGRATIO,
INCDENS.

(z) PCT RATIO UPPER is the upper limit of the percentile interval for the ex-
ponentiated treatment effect for TRANSFORM = LOGISTIC, PODDS, LOGRATIO,
INCDENS.



Chapter 2 Advanced Randomization-based Methods 99

6. OUTDAT HOMOGEN.

(a) TYPE = HOMOGEN.

(b) TRT1 is the value for control (lower value numerically).

(c) TRT2 is the value for treatment (higher value numerically).

(d) STRATA contains the strata variable.

(e) COVARIATES contains the list of covariates.

(f) OUTCOMES contains the list of outcomes.

(g) DF is the degrees of freedom.

(h) BETA is the treatment estimate.

(i) Q C is the test statistic.

(j) PVALUE is the p-value.

7. OUTDAT BETASAMP.

(a) FLAG = OBSERVED, PERMUTATION, BOOTSTRAP, JACKKNIFE for type of resam-
pled treatment estimate.

(b) BETASAMP is the treatment estimate.

(c) EXPBETASAMP is the exponentiated treatment estimate for TRANSFORM =

LOGISTIC, PODDS, LOGRATIO, INCDENS.

8. OUTDAT SURV.

(a) Same as the input data set but adds survival scores according to TRANSFORM.
For example, a variable called VAR will have one or the other of LOGRANK VAR

or WILCOXON VAR added to the data set.

9. OUTDAT COVBETA.

(a) TYPE = COVBETA.

(b) Other variables correspond to outcomes and covariates to identify rows and
columns of the covariance matrix.

10. OUTDAT STRATABETA.

(a) STRATA H contains the value for a strata level.

(b) BETA H contains the strata-specific treatment estimate.

(c) N H contains the sample size for a strata level.

